

```

MODEL:                ! ( QRinventory);
! Find the order quantity, Q,
  and re-order point, R, for a product with...;
DATA:
! In this self-contained example, data are hard coded in this file.
! In regular use you may wish to take data from an outside file.
! You can read data from an Excel file, and send results
  back to an Excel file with the @OLE() function;
!Case01 D = 280000;    ! Mean demand / year;
!Case01 H = 100;      ! Holding cost/unit/year;
!Case01 K = 400;      ! Fixed order cost;
!Case01 P = 200;      ! Penalty cost/unit unsatisfied demand;
!Case01 L = .0962;    ! Lead time in years;
!Case01 SDL = .03846; ! S.D. in lead time in years;
!Case01 SDD = 22000;  ! S.D. in yearly demand;

!Case02; D = 280000;  ! Mean demand / year;
!Case02; H = 100;     ! Holding cost/unit/year;
!Case02; K = 400;     ! Fixed order cost;
!Case02; P = 200;     ! Penalty cost/unit unsatisfied demand;
!Case02; L = .02;     ! Lead time in years;
!Case02; SDL = 0.001; ! S.D. in lead time in years;
!Case02; SDD = 22000; ! S.D. in yearly demand;
ENDDATA
! Keywords: @PSL, @PSN, Backlogging, Backorders, Inventory, LINGO,
  Linear loss function, Lost sales, Normal distribution,
  Periodic review, QR model, Safety stock;
!-----;
SUBMODEL QRmodel:
! The Q,R inventory model;
MLD = L * D;      ! Mean lead time demand;
! s.d. in lead time demand;
SLD = ( SDD * SDD * L + MLD * MLD * SDL * SDL)^.5;
! Expected cost/ period is ECOST;
MIN = ECOST;
ECOST = COSTORD + COSTCYC + COSTSFT + COSTPEN;
  COSTORD = ( K * D/ Q); ! Order cost per year;
  COSTCYC = H * Q/2;    ! Cost of holding cycle inventory;
  COSTSFT = H*( R - MLD + BR); ! Cost of holding safety stock;
  COSTPEN = P * D * BR/ Q; ! Cost penalty of unsatisfied demand;
! Expected amount short/cycle. @PSL() is
  the standard Normal linear loss function;
BR = SLD * @PSL( Z);
! @PSN() is the standard Normal left tail prob.;
@PSN( Z) = P * D / ( P * D + H * Q);
R = MLD + SLD * Z;      ! Reorder point;
@FREE( ECOST); @FREE( R);
@FREE( COSTORD); @FREE( COSTCYC);
@FREE( COSTSFT); @FREE( COSTPEN);
@FREE( Z); @FREE( BR);
ENDSUBMODEL

CALC:
  @SOLVE( QRmodel); ! Solve QR inventory model;
  @WRITE( @NEWLINE(1));
  @WRITE( @FORMAT( Q, ' 9.2f'), ' : Order quantity for each order.', @NEWLINE( 1));
  @WRITE( @FORMAT( R, ' 9.2f'), ' : Reorder point to trigger an order.', @NEWLINE( 1));
ENDCALC
END

```